Representations of Piecewise Continuous Functions

Univ.-Prof. Dr.-Ing. habil. Josef BETTEN
RWTH Aachen University
Templergraben 55
D-52056 A a c h e n , Germany

<betten@mmw.rwth-aachen.de>

Abstract

This worksheet is concerned with representations of piecewise continuos functions. It has been illustrated that Maplesoft furnishes powerful tools in finding such functions by using the command < piecewise > or, alternatively, by introducing *HEAVISIDE* functions. Some examples are discussed in more detail.

Keywords: FÖPPL; MACAULY; HEAVISIDE; DIRAC

Introduction

In solving beam problems, for instance, representations of piecewise continuous loading functions play a fundamental role. Loads commonly applied to a beam may consist of concentrated forces, applied at a point, or piecewise continuously distributed loads.
Compactness of representation may be achieved by introducing *singularity* or *half-range* functions. Such functions were applied to bending proplems by *MACAULEY* in 1919 or by *FÖPPL* (1854 - 1924).

***FÖPPL*-Symbol**

The so-called *FÖPPL*-symbol is indicated by pointed brackets < x - a > and defined to be zero if x < a and to be simply (x - a) if x > a :

```
> restart:
> FÖPPL_bracked:=piecewise(x<a,0, x>a,(x-a));
```

$$FÖPPL\_bracked := \begin{cases} 0 & x < a \\ x - a & a < x \end{cases}$$

especially we have:

```
> F(x,a,n):=piecewise(x<a,0, x>a,(x-a)^n);
```

$$F(x, a, n) := \begin{cases} 0 & x < a \\ (x-a)^n & a < x \end{cases}$$

where the differentiation and integration are given as:

```
> Derivative(x):=Diff(F(x),x)=diff(F(x,a,n),x);
```

$$\mathrm{Derivative}(x) := \frac{d}{dx} F(x) = \begin{cases} 0 & x < a \\ \mathit{undefined} & x = a \\ \dfrac{(x-a)^n\, n}{x-a} & a < x \end{cases}$$

```
> Indefinite_Integral(x):=Int(F(x),x)=int(F(x,a,n),x);
```

$$\mathrm{Indefinite\_Integral}(x) := \int F(x)\, dx = \begin{cases} 0 & x \le a \\ \dfrac{(x-a)^{(n+1)}}{n+1} & a < x \end{cases}$$

The above piecewise function due to *FÖPPL* or *MACAULEY* can also be represented by the *HEAVISIDE*-function:

```
> alias(H=Heaviside):
> HEAVISIDE[F]:=convert(F(x,a,n),H);
```

$$HEAVISIDE_F := (x-a)^n\, \mathrm{H}(x-a)$$

where:

```
> H(x-a):=piecewise(x<a,0, x>a,1);
```

$$\mathrm{H}(x-a) := \begin{cases} 0 & x < a \\ 1 & a < x \end{cases}$$

```
> Derivative[HEAVISIDE]:=Diff(H,x)=Dirac(x-a);
```

$$Derivative_{HEAVISIDE} := \frac{\partial}{\partial x} H = \mathrm{Dirac}(x-a)$$

This function is called the *DIRAC delta function* defined as:

```
> delta(x-a):=piecewise(x<a and x>a,0, x=a,infinity);
```

$$\delta(x-a) := \begin{cases} 0 & x < a \textbf{ and } a < x \\ \infty & x = a \end{cases}$$

the properties of which are given by:

```
> Int(delta(x - alpha),x=0..infinity)=
  int(Dirac(x-a),x=0..infinity);
```

$$\int_0^\infty \delta(x-\alpha)\, dx = \mathrm{H}(a)$$

where

```
> Heaviside(a):=1;
```

$$\mathrm{H}(a) := 1$$

for arbitrary real values of $a > 0$. Thus, we find:

```
> Itegration[0..infinity]:=
  Int(delta(x - alpha)*f(x),x=0..infinity)=f(alpha);
```

$$Itegration_{0\,..\,\infty} := \int_0^\infty \delta(x-\alpha)\, \mathrm{f}(x)\, dx = \mathrm{f}(\alpha)$$

```
>
```

Unit Step Function or *HEAVISIDE* Unit Function

```
> restart:
> F(x):=piecewise(x>0 and x<Pi,1,x>Pi and x<2*Pi,-1,x>2*Pi,1);
```

$$F(x) := \begin{cases} 1 & 0 < x \text{ and } x < \pi \\ -1 & \pi < x \text{ and } x < 2\pi \\ 1 & 2\pi < x \end{cases}$$

```
> alias(th=thickness,co=color):
> plot(F(x),x=0..2.5*Pi,th=3,labels=[x,F],co=black,
  title="Unit Step Function");
```



Alternatively, the piecewise continuous function F(x) can be represented as a *HEAVISIDE* function by using the command *convert*:

```
> alias(H=Heaviside):
> HEAVISIDE[F]:=convert(F(x),H);
```

$$HEAVISIDE_F := \mathrm{H}(x) - 2\,\mathrm{H}(x-\pi) + 2\,\mathrm{H}(x-2\pi)$$

```
> plot(%,x=0..2.5*Pi,th=3,co=black,
  title="HEAVISIDE Unit Step Function");
```




```
>
```

The next example is concerned with the
FÖPPL-function < x - a > := x = 0 for x < a and (x - a)^n for x > a

```
> restart:
> F(x,a,n):=piecewise(x>0 and x<a,0, x>a,(x-a)^n);
```

$$F(x, a, n) := \begin{cases} 0 & 0 < x \text{ and } x < a \\ (x-a)^n & a < x \end{cases}$$

```
> F(x,1/3,1/2):=subs({a=1/3,n=1/2},F(x,a,n));
```

$$F\left(x, \frac{1}{3}, \frac{1}{2}\right) := \begin{cases} 0 & 0 < x \text{ and } x < \frac{1}{3} \\ \sqrt{x - \frac{1}{3}} & \frac{1}{3} < x \end{cases}$$

```
> alias(sc=scaling,th=thickness,co=color):
> plot(F(x,1/3,1/2),x=0..1,sc=constrained,th=3,co=black);
```




```
> Integral(1/3,1/2):=Int(F(x),x=0..1)=int(F(x,1/3,1/2),x=0..1);
```

$$\text{Integral}\left(\frac{1}{3}, \frac{1}{2}\right) := \int_0^1 F(x)\, dx = \frac{4\sqrt{6}}{27}$$

```
> Integral(1/3,1/2):=evalf(rhs(%),5);
```

$$\text{Integral}\left(\frac{1}{3}, \frac{1}{2}\right) := 0.36289$$

```
> Derivative(x,a,n):=Diff(F(x),x)=diff(F(x,a,n),x);
```

$$\text{Derivative}(x, a, n) := \frac{d}{dx} F(x) = \begin{cases} 0 & 0 < x \text{ and } x < a \\ \frac{(x-a)^n\, n}{x - a} & a < x \end{cases}$$

```
> Derivative(x,1/3,1/2):=subs({a=1/3,n=1/2},%);
```

$$\text{Derivative}\left(x, \frac{1}{3}, \frac{1}{2}\right) := \frac{d}{dx} F(x) = \begin{cases} 0 & 0 < x \text{ and } x < \frac{1}{3} \\ \frac{1}{2\sqrt{x - \frac{1}{3}}} & \frac{1}{3} < x \end{cases}$$

```
> plot(rhs(%),x=0..1,0..5,th=3,co=black);
```




```
>
```

The same result can be achieved by using a *HEAVISIDE* function:

```
> alias(H=Heaviside):
> G(x,a,n):=convert(F(x,a,n),H);
```

$$G(x, a, n) := (x-a)^n \, H(x-a)$$

```
> G(x,1/3,1/2):=subs({a=1/3,n=1/2},G(x,a,n));
```

$$G\left(x, \frac{1}{3}, \frac{1}{2}\right) := \sqrt{x-\frac{1}{3}}\, H\left(x-\frac{1}{3}\right)$$

```
> plot(%,x=0..1,sc=constrained,th=3,co=black);
```




```
> INTEGRAL(1/3,1/2):=Int(G(x),x=0..1)=int(G(x,1/3,1/2),x=0..1);
```

$$\mathrm{INTEGRAL}\left(\frac{1}{3}, \frac{1}{2}\right) := \int_0^1 G(x)\, dx = \frac{4\sqrt{6}}{27}$$

```
> INTEGRAL(1/3,1/2):=evalf(rhs(%),5);
```

$$\mathrm{INTEGRAL}\left(\frac{1}{3}, \frac{1}{2}\right) := 0.36289$$

```
> DERIVATIVE(x,a,n):=Diff(G(x),x)=diff(G(x,a,n),x);
```

$$\mathrm{DERIVATIVE}(x, a, n) := \frac{d}{dx} G(x) = \frac{(x-a)^n \, n \, H(x-a)}{x-a} + (x-a)^n \, \mathrm{Dirac}(x-a)$$

```
> DERIVATIVE(x,1/3,1/2):=subs({a=1/3,n=1/2},%);
```

$$\mathrm{DERIVATIVE}\left(x, \frac{1}{3}, \frac{1}{2}\right) := \frac{d}{dx}\mathrm{G}(x) = \frac{1}{2}\frac{\mathrm{H}\left(x-\frac{1}{3}\right)}{\sqrt{x-\frac{1}{3}}} + \sqrt{x-\frac{1}{3}}\,\mathrm{Dirac}\left(x-\frac{1}{3}\right)$$

```
> plot(rhs(%),x=0..1,0..5,th=3,co=black);
```




```
>
```

For n = 0 we arrive at:

```
> F(x,a,0):=subs(n=0,F(x,a,n));
```

$$\mathrm{F}(x, a, 0) := \begin{cases} 0 & 0 < x \textbf{ and } x < a \\ 1 & a < x \end{cases}$$

```
> F(x,1/3,0):=subs(a=1/3,%);
```

$$\mathrm{F}\left(x, \frac{1}{3}, 0\right) := \begin{cases} 0 & 0 < x \textbf{ and } x < \frac{1}{3} \\ 1 & \frac{1}{3} < x \end{cases}$$

```
> plot(F(x,1/3,0),x=0..1,sc=constrained,th=3,co=black);
```




```
> G(x,1/3,0):=convert(F(x,1/3,0),H);
```

$$\mathrm{G}\left(x, \frac{1}{3}, 0\right) := \mathrm{H}\left(x-\frac{1}{3}\right)$$

```
> plot(%,x=0..1,sc=constrained,th=3,co=black);
```




```
>
```

Concentrated force F = 1/4 on a beam at x = a =1/3

```
> restart:
> f(x,a,F):=piecewise(x>0 and x<a,0, x=a,F, x>a,0);
```

$$f(x, a, F) := \begin{cases} 0 & 0 < x \textbf{ and } x < a \\ F & x = a \\ 0 & a < x \end{cases}$$

```
> f(x,1/3,1/4):=subs({a=1/3,F=1/4},%);
```

$$f\left(x, \frac{1}{3}, \frac{1}{4}\right) := \begin{cases} 0 & 0 < x \textbf{ and } x < \frac{1}{3} \\ \frac{1}{4} & x = \frac{1}{3} \\ 0 & \frac{1}{3} < x \end{cases}$$

```
>
```

Alternatives: *HEAVISIDE* **function** / **impulse** or *DIRAC* **delta function**

```
> alias(H=Heaviside,th=thickness,co=color):
> g(x,a,F):=convert(f(x,a,F),H);
```

$$g(x, a, F) := F \operatorname{Dirac}(x - a)$$

```
> Integral:=Int(g(x),x=0..infinity)=
  int(g(x,a,F),x=0..infinity);
```

$$Integral := \int_0^\infty g(x)\, dx = F \operatorname{H}(a)$$

```
> H(a):=H(1/4)=H(5);  # for every value "a > 0"
```

$$\operatorname{H}(a) := 1 = 1$$

```
> H(b):=H(-1/4)=H(-5); # for every value  "b < 0"
```

$$\operatorname{H}(b) := 0 = 0$$

H(a) is not defined for $a = 0$ and for non-real values in HEAVISIDE.

```
> g(x,1/3,1/4):=subs({a=1/3,F=1/4},g(x,a,F));
```

$$g\left(x, \frac{1}{3}, \frac{1}{4}\right) := \frac{1}{4} \operatorname{Dirac}\left(x - \frac{1}{3}\right)$$

```
> Int(g(x),x=0..infinity)=int(g(x,1/3,1/4),x=0..infinity);
```

$$\int_0^{\infty} g(x)\,dx = \frac{1}{4}$$

```
> Derivative(x,a):=Diff(H(x-a),x)=diff(H(x-a),x);
```

$$\text{Derivative}(x, a) := \frac{\partial}{\partial x} \text{H}(x - a) = \text{Dirac}(x - a)$$

```
> Derivative(x,1/3):=subs(a=1/3,rhs(%));
```

$$\text{Derivative}\left(x, \frac{1}{3}\right) := \text{Dirac}\left(x - \frac{1}{3}\right)$$

```
> G(x,a,F):=F*H(x-a)-F*H(x-1.005*a);
```

$$\text{G}(x, a, F) := F\,\text{H}(x - a) - F\,\text{H}(x - 1.005\,a)$$

```
> Derivative(x,a,F):=Diff(G(x),x)=diff(%,x);
```

$$\text{Derivative}(x, a, F) := \frac{d}{dx} \text{G}(x) = F\,\text{Dirac}(x - a) - F\,\text{Dirac}(x - 1.005\,a)$$

```
> G(x,1/3,1/4):=subs({a=1/3,F=1/4},G(x,a,F));
```

$$\text{G}\left(x, \frac{1}{3}, \frac{1}{4}\right) := \frac{1}{4} \text{H}\left(x - \frac{1}{3}\right) - \frac{1}{4} \text{H}(x - 0.3350000000)$$

```
> plot(%,x=0..1,th=3,co=black,labels=[x,F],
  title="Single  Force  generated  by  HEAVISIDE");
```




```
>
```

The single force F = 1/4 at x = 1/3 in this Figure can alternatively be represented by using the *signum function* defined as *csgn* in Maple:

```
> restart:
> signum(x):=csgn(x)=piecewise(x<0,-1, x=0,0, x>0,1);
```

$$\text{signum}(x) := \text{csgn}(x) = \begin{cases} -1 & x < 0 \\ 0 & x = 0 \\ 1 & 0 < x \end{cases}$$

```
> signum(x-a):=csgn(x-a)=piecewise(x<a,-1, x=a,0, x>a,1);
```

$$\text{signum}(x - a) := -\text{csgn}(-x + a) = \begin{cases} -1 & x < a \\ 0 & x = a \\ 1 & a < x \end{cases}$$

```
> alias(th=thickness,co=color):
> plot(csgn(x-1),x=-1..3,th=3,co=black,
  title="Signum  Function  csgn(x-1)");
```




```
> p[1]:=plot((1/4)*csgn(x-1/3),x=1/3.001..1/2.99,0..1/4,
th=3,co=black,labels=[x,F]):
> p[2]:=plot(0,x=0..1,
title="Single  Force generated  by  Signum  Function"):
> plots[display](seq(p[k],k=1..2));
```




```
>
```

The next example represents the following piecewise continuous function:

```
> restart:
> F(x):=piecewise(-Pi<x and x<0,1, x>0 and x<1/2,2,
x>1/2 and x<Pi,1/x);
```

$$\mathrm{F}(x) := \begin{cases} 1 & -\pi < x \textbf{ and } x < 0 \\ 2 & 0 < x \textbf{ and } x < \frac{1}{2} \\ \frac{1}{x} & \frac{1}{2} < x \textbf{ and } x < \pi \end{cases}$$

```
> alias(H=Heaviside,th=thickness,co=color):
> plot(F(x),x=-Pi..Pi,th=3,co=black,labels=[x,F]);
```




```
> Integral[-Pi..Pi]:=Int(F,x=-Pi..Pi)=int(F(x),x=-Pi..Pi);
```

$$Integral_{-\pi\,..\,\pi} := \int_{-\pi}^{\pi} F\,dx = 1 + \pi + \ln(2) + \ln(\pi)$$

```
> Integral[-Pi..Pi]:=evalf(rhs(%),5);
```

$$Integral_{-\pi\,..\,\pi} := 5.9795$$

```
> Derivative(x):=Diff(F,x)=diff(F(x),x);
```

$$\mathrm{Derivative}(x) := \frac{\partial}{\partial x} F = \begin{cases} 0 & x < -\pi \\ undefined & x = -\pi \\ 0 & x < 0 \\ undefined & x = 0 \\ 0 & x < \frac{1}{2} \\ undefined & x = \frac{1}{2} \\ -\frac{1}{x^2} & x < \pi \\ undefined & x = \pi \\ 0 & \pi < x \end{cases}$$

```
> plot(rhs(%),x=-Pi..Pi,-Pi..0,th=3,co=black);
```




```
>
> HEAVISIDE[F]:=convert(F(x),H);
```

$$HEAVISIDE_F := \mathrm{H}(x+\pi) + \mathrm{H}(x) - 2\,\mathrm{H}\left(x - \frac{1}{2}\right) + \frac{\mathrm{H}\left(x - \frac{1}{2}\right)}{x} - \frac{\mathrm{H}(x-\pi)}{x}$$

```
> INTEGRAL[-Pi..Pi]:=Int(HEAVISIDE,x=-Pi..Pi)=int(%,x=-Pi..Pi);
```

$$INTEGRAL_{-\pi..\pi} := \int_{-\pi}^{\pi} HEAVISIDE\, dx = \pi + 1 + \ln(\pi) + \ln(2)$$

```
> INTEGRAL[-Pi..Pi]:=evalf(rhs(%),5);
```

$$INTEGRAL_{-\pi..\pi} := 5.9794$$

```
> DERIVATIVE(x):=Diff(HEAVISIDE,x)=diff(HEAVISIDE[F],x);
```

$$\mathrm{DERIVATIVE}(x) := \frac{\partial}{\partial x} HEAVISIDE = \mathrm{Dirac}(x+\pi) + \mathrm{Dirac}(x) - 2\,\mathrm{Dirac}\left(x - \frac{1}{2}\right) - \frac{\mathrm{H}\left(x - \frac{1}{2}\right)}{x^2} + \frac{\mathrm{Dirac}\left(x - \frac{1}{2}\right)}{x} + \frac{\mathrm{H}(x-\pi)}{x^2} - \frac{\mathrm{Dirac}(x-\pi)}{x}$$

```
> plot(rhs(%),x=-Pi..Pi,-Pi..0,th=3,co=black);
```




```
>
```

The piecewise function h(x) is defined as follows:

```
> restart:
> h(x):=piecewise(x>0 and x<Pi,Pi, x>Pi and x<2*Pi,x-Pi,
  x>2*Pi and x<3*Pi,Pi);
```

$$\mathrm{h}(x) := \begin{cases} \pi & 0 < x \textbf{ and } x < \pi \\ x - \pi & \pi < x \textbf{ and } x < 2\pi \\ \pi & 2\pi < x \textbf{ and } x < 3\pi \end{cases}$$

```
> alias(H=Heaviside,th=thickness,co=color):
> plot(h(x),x=0..3*Pi,th=3,co=black,labels=[x,h],
  title="Piecewise  Continuous  Function  h(x)");
```




```
> Integral[0..3*Pi]:=Int(h,x=0..3*Pi)=int(h(x),x=0..3*Pi);
```

$$Integral_{0..3\pi} := \int_0^{3\pi} h\, dx = \frac{5\pi^2}{2}$$

```
> Integral[0..3*Pi]:=evalf(rhs(%),5);
```

$$Integral_{0..3\pi} := 24.674$$

```
> Derivative(x):=Diff(h,x)=diff(h(x),x);
```

$$\text{Derivative}(x) := \frac{\partial}{\partial x} h = \begin{cases} 0 & x < 0 \\ undefined & x = 0 \\ 0 & x < \pi \\ undefined & x = \pi \\ 1 & x < 2\pi \\ undefined & x = 2\pi \\ 0 & x < 3\pi \\ undefined & x = 3\pi \\ 0 & 3\pi < x \end{cases}$$

```
> plot(rhs(%),x=0..3*Pi,th=3,co=black,
  title="Derivative  of  h(x)");
```




```
> HEAVISIDE[h]:=convert(h(x),H);
```

$$HEAVISIDE_h :=$$

$$\pi\, \mathrm{H}(x) - 2\pi\, \mathrm{H}(x-\pi) + x\, \mathrm{H}(x-\pi) - x\, \mathrm{H}(x-2\pi) + 2\pi\, \mathrm{H}(x-2\pi) - \pi\, \mathrm{H}(x-3\pi)$$

```
> plot(%,x=0..3*Pi,th=3,co=black,
  title="HEAVISIDE  Function  for  h(x)");
```




```
> INTEGRAL[0..3*Pi]:=Int(HEAVISIDE,x=0..3*Pi)=
  int(HEAVISIDE[h],x=0..3*Pi);
```

$$INTEGRAL_{0..3\pi} := \int_0^{3\pi} HEAVISIDE\, dx = \frac{5\pi^2}{2}$$

```
> INTEGRAL[0..3*Pi]:=evalf(rhs(%),5);
```

$$INTEGRAL_{0..3\pi} := 24.674$$

```
> DERIVATIVE(x):=Diff(HEAVISIDE,x)=diff(HEAVISIDE[h],x);
```

$$\mathrm{DERIVATIVE}(x) := \frac{\partial}{\partial x} HEAVISIDE = \pi\,\mathrm{Dirac}(x) - 2\,\pi\,\mathrm{Dirac}(x-\pi) + \mathrm{H}(x-\pi) + x\,\mathrm{Dirac}(x-\pi) - \mathrm{H}(x-2\,\pi) - x\,\mathrm{Dirac}(x-2\,\pi) + 2\,\pi\,\mathrm{Dirac}(x-2\,\pi) - \pi\,\mathrm{Dirac}(x-3\,\pi)$$

```
> plot(rhs(%),x=0..3*Pi,th=3,co=black,
  title="Derivative  of  HEAVISIDE[h]");
```




```
>
```

Discontinuous Function:

```
> restart:
> F(x):=piecewise(x>-Pi and x<0,3, x>1/2 and x<Pi,1/x);
```

$$F(x) := \begin{cases} 3 & -\pi < x \textbf{ and } x < 0 \\ \dfrac{1}{x} & \dfrac{1}{2} < x \textbf{ and } x < \pi \end{cases}$$

```
> alias(th=thickness,co=color):
> plot(F(x),x=-Pi..Pi,th=3,co=black,labels=[x,F]);
```




```
> Integral[-Pi..Pi]:=Int(F,x=-Pi..Pi)=int(F(x),x=-Pi..Pi);
```

$$Integral_{-\pi..\pi} := \int_{-\pi}^{\pi} F\,dx = 3\,\pi + \ln(2) + \ln(\pi)$$

```
> Integral[-Pi..Pi]:=evalf(rhs(%),5);
```

$$Integral_{-\pi..\pi} := 11.263$$

```
> Derivative(x):=Diff(F,x)=diff(F(x),x);
```

$$\mathrm{Derivative}(x) := \frac{\partial}{\partial x} F = \begin{cases} 0 & x < -\pi \\ \mathit{undefined} & x = -\pi \\ 0 & x < 0 \\ \mathit{undefined} & x = 0 \\ 0 & x < \frac{1}{2} \\ \mathit{undefined} & x = \frac{1}{2} \\ -\frac{1}{x^2} & x < \pi \\ \mathit{undefined} & x = \pi \\ 0 & \pi < x \end{cases}$$

```
> plot(rhs(%),x=-Pi..Pi,-Pi..0,th=3,co=black);
```




```
>
> alias(H=Heaviside):
> G(x):=convert(F(x),H);
```

$$\mathrm{G}(x) := 3\,\mathrm{H}(x+\pi) - 3\,\mathrm{H}(x) + \frac{\mathrm{H}\left(x - \frac{1}{2}\right)}{x} - \frac{\mathrm{H}(x-\pi)}{x}$$

```
> plot(G(x),x=-Pi..Pi,th=3,co=black,labels=[x,G]);
```




```
> INTEGRAL[-Pi..Pi]:=Int(G,x=-Pi..Pi)=int(G(x),x=-Pi..Pi);
```

$$\mathit{INTEGRAL}_{-\pi\,..\,\pi} := \int_{-\pi}^{\pi} G\,dx = 3\,\pi + \ln(2) + \ln(\pi)$$

```
> INTEGRAL[-Pi..Pi]:=evalf(rhs(%),5);
```

$$\mathit{INTEGRAL}_{-\pi\,..\,\pi} := 11.263$$

```
> DERIVATIVE(x):=Diff(G,x)=diff(G(x),x);
```

DERIVATIVE(x) :=

$$\frac{\partial}{\partial x} G = 3\,\mathrm{Dirac}(x+\pi) - 3\,\mathrm{Dirac}(x) - \frac{\mathrm{H}\left(x-\frac{1}{2}\right)}{x^2} + \frac{\mathrm{Dirac}\left(x-\frac{1}{2}\right)}{x} + \frac{\mathrm{H}(x-\pi)}{x^2} - \frac{\mathrm{Dirac}(x-\pi)}{x}$$

```
> plot(rhs(%),x=-Pi..Pi,-Pi..0,th=3,co=black);
```



Further picewise continuous functions and their corresponding *LAPLACE* transformation have been applied to creep problems by *BETTEN, J.* (2008) in "Creep Mechanics, Third Edition, Springer-Verlag, Berlin / Heidelberg / New York.

Modified *DIRAC* Delta Function

As has already been mentioned above the *DIRAC delta function* is defined as follows:

```
> restart:
> alias(H=Heaviside):
> delta(x-a):=piecewise(x<a and x>a,0, x=a,infinity);
```

$$\delta(x-a) := \begin{cases} 0 & x < a \text{ and } a < x \\ \infty & x = a \end{cases}$$

the properties of which are given by:

```
> Int(delta(x-alpha),x=0..infinity)=
  int(Dirac(x-a),x=0..infinity);
```

$$\int_0^\infty \delta(x-\alpha)\,dx = \mathrm{H}(a)$$

and

```
> Int(delta(x-alpha)*f(x),x=0..infinity)=
  int(Dirac(x-a)*f(x),x=0..infinity);
```

$$\int_0^\infty \delta(x-\alpha)\,\mathrm{f}(x)\,dx = \mathrm{H}(a)\,\mathrm{f}(a)$$

where

```
> HEAVISIDE(a>0):=H(a) assuming a > 0;
```

$$\mathrm{HEAVISIDE}(0 < a) := 1$$

```
> HEAVISIDE(a<0):=H(a) assuming a < 0;
```

$$\mathrm{HEAVISIDE}(a < 0) := 0$$

```
> HEAVISIDE(a=0):=H(a) assuming a = 0;
```

$$\text{HEAVISIDE}(a=0) := \textit{undefined}$$

```
> HEAVISIDE(a=I):=H(I);
Error, (in Heaviside) not defined for non-real values
```

H(a) is not defined for a = 0 and for non-real values in HEAVISIDE .

The first property of the delta function can geometrically interpreted as illustrated in the following Figure representing a rectangular impulse of width "epsilon" and height "1/epsilon" at a point x = a .

```
> restart:
> alias(th=thickness,co=color):
> plot(piecewise(x>0 and x<3/2,0, x>3/2 and x<3/2+1/2,2,
  x>3/2+1/2,0), x=0..3,th=3,co=black,labels=[x,DIRAC[modified]]);
```




```
>
```

Where the modified *DIRAC* delta function is defined as follows:

```
> delta[epsilon](x-a):=piecewise(x>0 and x<a,0,
  x>a and x<a+epsilon,1/epsilon, x>a+epsilon,0);
```

$$\delta_{\varepsilon}(x-a) := \begin{cases} 0 & 0 < x \textbf{ and } x < a \\ \dfrac{1}{\varepsilon} & a < x \textbf{ and } x < a+\varepsilon \\ 0 & a+\varepsilon < x \end{cases}$$

tending to *DIRAC* as epsilon tends to zero, where the rectangular area in the Figure is always unity: epsilon * (1/epsilon). Thus, we have:

```
> area:=Int(delta[epsilon](x-alpha),x=a..a+epsilon)=
  int(1/epsilon,x=a..a+epsilon);
```

$$\textit{area} := \int_{a}^{a+\varepsilon} \delta_{\varepsilon}(x-\alpha)\, dx = 1$$

Hence, the first property of the *DIRAC* function is valid.

If any continuous function f(x) is multiplied by the *DIRAC* delta function the product vanishes everywhere except at x = a , while the value f(a) is independent of x and can therefore be written outside the integral. Thus, considering the first property, we arrive at the above second property of the *DIRAC* function, which can also be applied to the *collocation method* according to:

```
> R(X[i]):=Int(delta(x-x[i])*R(x),x=0..L)=0;
```

$$R(X_i) := \int_0^L \delta(x - x_i)\, R(x)\, dx = 0$$

where R(x) is the *Residuum* of an approximation and X[i] is an optimal selected *collocation point.*
The integral is setting over a domain L considered. The *collocation method* can be shown to be a special case of the *weighted-residual method:*

```
> Int(W[i](x)*R(x),x=0..L)=0;
```

$$\int_0^L W_i(x)\, R(x)\, dx = 0$$

where W[i](x) are *weight functions* [*BETTEN, J.* (2004), Finite Elemente für Ingenieure 2, zweite Aufl., Springer-Verlag, Berlin / Heidelberg / New York]. Further examples concerning modified *DIRAC* and modified *HEAVISIDE* functions have been discussed in more detail by *BETTEN, J.* (2008) in "Creep Mechanics, Third Edition, Springer-Verlag ".

```
>
```